

Numerical Representation of Planetary Ephemerides

X. X. Newhall, Celestial Mechanics 45:305-310, 1989

Computations

A handy function to compute the derivative of a Chebyshev polynomial.

```
DChebyshevT = Derivative[0, 1][ChebyshevT]
```

```
ChebyshevU[-1 + #1, #2] #1 &
```

This function computes matrix T from Newhall's equation (5). The parameter *degree* is the degree of the polynomial (N in Newhall), the parameter *divisions* is the number of subintervals of $[-1, 1]$ (8 in Newhall).

```
NewhallT[degree_Integer, divisions_Integer] :=  
  Flatten[  
    Table[  
      {  
        Table[ChebyshevT[j, i], {j, 0, degree}],  
        Table[DChebyshevT[j, i], {j, 0, degree}]  
      },  
      {i, 1, -1, -2 / divisions}  
    ], {1, 2}]
```

This function computes matrix W used in Newhall's equation (8). The parameter w is the weight of the velocities relative to the positions (0.4 in Newhall).

```
NewhallW[divisions_Integer, w_Rational] :=  
  DiagonalMatrix[Flatten[Table[{1, w^2}, {divisions + 1}]]]
```

The following functions compute the four blocks of matrix $C1$ and assemble them to form $C1$.

```
NewhallC1UpperLeft[degree_Integer, divisions_Integer, w_Rational] :=  
  NewhallT[degree, divisions]^T.NewhallW[divisions, w].NewhallT[degree, divisions]
```

```
NewhallC1UpperRight[degree_Integer] :=  
  Table[  
    {ChebyshevT[i, 1], DChebyshevT[i, 1],  
     ChebyshevT[i, -1], DChebyshevT[i, -1]}, {i, 0, degree}  
  ]
```

```
NewhallC1LowerLeft[degree_Integer] := NewhallC1UpperRight[degree]^T
```

```
NewhallC1LowerRight[] := Table[0, {4}, {4}]
```

```
NewhallC1[degree_Integer, divisions_Integer, w_Rational] := ArrayFlatten[  
  {{NewhallC1UpperLeft[degree, divisions, w], NewhallC1UpperRight[degree]},  
   {NewhallC1LowerLeft[degree], NewhallC1LowerRight[]}}  
]
```

The following functions compute the two blocs of matrix $C2$ and assemble them to form $C2$.

```

NewhallC2Upper[degree_Integer, divisions_Integer, w_Rational] :=
  NewhallT[degree, divisions]^T.NewhallW[divisions, w]
NewhallC2Lower[divisions_Integer] :=
  Drop[IdentityMatrix[2 divisions + 2], {3, 2 divisions}]
NewhallC2[degree_Integer, divisions_Integer, w_Rational] :=
  ArrayFlatten[{{NewhallC2Upper[degree, divisions, w]}, {NewhallC2Lower[divisions]}}]

```

This function computes the matrix $C1^{-1}.C2$. Newhall doesn't give it a name but calls its elements c_k , so let's use the name C .

```

NewhallC[degree_Integer, divisions_Integer, w_Rational] :=
  Inverse[NewhallC1[degree, divisions, w]].NewhallC2[degree, divisions, w]

```

Formatting and Output

Produces a representation of a matrix as an initializer_list containing initializer_lists.

```

BidimMatrixToCDefinition[type_String, variable_String, matrix_List] :=
  type <> " const\r\n      " <> variable <> "(\r\n" <>
  StringReplace[
    ToString[
      CForm[matrix]
    ],
    {
      "List(List(" → "      {",
      "List(" → "{",
      ")," → "},\r\n      ",
      ", " → ",\r\n      ",
      ")))" → "});\r\n\r\n"
    }
  ]

```

Produces a representation of a matrix as a single, flattened initializer list.

```

FlattenedMatrixToCDefinition[type_String, variable_String, matrix_List] :=
  type <> " const\r\n      " <> variable <> "(\r\n" <>
  StringReplace[
    ToString[
      CForm[matrix]
    ],
    {
      "List(List(" → "      {",
      "List(" → "\r\n      ",
      ")," → ",\r\n",
      ", " → ",\r\n      ",
      ")))" → "});\r\n\r\n"
    }
  ]

```

Produces a representation of a list as an initializer list.

```
ListToCDefinition[type_String, variable_String, list_List] :=
  type <> " const\r\n      " <> variable <> "(\r\n" <>
  StringReplace[
    ToString[
      CForm[list]
    ],
    {
      "List(" → "      {",
      ", " → ",\r\n      ",
      ")" → "});\r\n\r\n"
    }
  ]
]
```

Writes all the Newhall *C* matrices to a single file. Note that we drop the last 4 rows because they correspond to the Lagrange multipliers.

```
file = OpenWrite[FileNameJoin[{DirectoryName[NotebookDirectory[]], "numerics",
  "newhall.mathematica.cpp"}], BinaryFormat → True, PageWidth → Infinity];
WriteString[
  file,
  "#pragma once\r\n",
  "\r\n",
  "#include <array>\r\n",
  "\r\n",
  "#include \"numerics/arrays.hpp\"\r\n",
  "\r\n",
  "namespace principia {\r\n",
  "namespace numerics {\r\n",
  ];
Do[
  WriteString[
    file,
    FlattenedMatrixToCDefinition[
      "FixedMatrix<double, " <> ToString[degree] <> " + 1, 2 * 8 + 2>",
      "newhall_c_matrix_degree_" <> ToString[degree] <> "_divisions_8_w04",
      Drop[NewhallC[degree, 8, 4 / 10], -4]
    ]
  ],
  {degree, 3, 17}];
WriteString[
  file,
  "} // namespace numerics\r\n",
  "} // namespace principia\r\n"
];
Close[file]
C:\Users\phl\Projects\GitHub\Principia\numerics\newhall.mathematica.cpp
```